# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=perl-test-yaml
pkgver=1.07
pkgrel=6
pkgdesc="Testing Module for YAML Implementations"
arch=('any')
url="https://search.cpan.org/dist/Test-YAML/"
license=('GPL' 'PerlArtistic')
depends=('perl-test-base')
options=('!emptydirs')
source=("https://www.cpan.org/authors/id/T/TI/TINITA/Test-YAML-${pkgver}.tar.gz")
sha512sums=('fe169d351e3c3e41c04d4429bc698f0aa49c7572e1681d4d72d5610d56234642093a31ed7ee5a972cac3e714ff481aa31e69bf12807ec446dd6f927e854d5606')

build() {
  cd Test-YAML-${pkgver}
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd Test-YAML-${pkgver}
  make test
}

package() {
  cd Test-YAML-${pkgver}
  make DESTDIR="${pkgdir}" install
}
